

# Modeling and analyzing the Corona-virus warning app with the Isabelle Infrastructure framework

Florian Kammüller and Bianca Lutz

Middlesex University London and  
Technische Universität Berlin

`f.kammueeller@mdx.ac.uk|bianca.lutz@campus.tu-berlin.de`

**Abstract.** We provide a model in the Isabelle Infrastructure framework of the recently published German Corona-virus warning app (CWA). The app supports breaking infection chains by informing users whether they have been in close contact to an infected person. The app has a decentralized architecture that supports anonymity of users. We provide a formal model of the existing app with the Isabelle Infrastructure framework to show up some natural attacks in a very abstract model. We then use the security refinement process of the Isabelle Infrastructure framework to highlight how the use of continuously changing Ephemeral Ids (EphIDs) improves the anonymity.

## 1 Introduction

The German Chancellor Angela Merkel has strongly supported the publication of the mobile phone Corona-virus warning app (CWA; [14]) by publicly proclaiming that “this App deserves your trust” [2]. Many millions of mobile phone users in Germany have downloaded the app with 6 million on the first day. CWA is one amongst many similar applications that aim at the very important goal to “break infection chains” by providing timely information to users of whether they have been in close proximity to a person who tested positive for COVID-19.

The app was designed with great attention on privacy: a distributed architecture [15] has been adopted that is based on a very clever application design whereby clients broadcast highly anonymized identifiers (ids) via Bluetooth and store those ids of people in close proximity. Infected persons report their infection by uploading their ids to a central server, providing all clients the means to assess exposure risk locally, hence, stored contact data has never to be shared.

The Isabelle Infrastructure framework [11] allows modeling and analyzing architecture and scenarios including physical and logical entities, actors, and policies within the interactive theorem prover Isabelle supported with temporal logic, Kripke structures, and attack trees. It has been applied for example to Insider analysis in airplanes [12], privacy in IoT healthcare [4], and recently also to blockchain protocols [10].

The technical advantage of modeling an application in the Isabelle Insider framework lies in (a) having explicit representations of infrastructures, actors and policies in a formal model that (b) allows additional automated verification of security properties with CTL, Kripke structures and Attack Trees within the interactive theorem prover Isabelle.

The app now in use in Germany has been developed based on a protocol proposed by the DP-3T project [16]: A sophisticated security concept conceived by experts in the field that has strong claims with regard to mathematical support [17] [p2]. However, there has, as of yet, to our knowledge no formal verification been involved. Even if a “post-production” formal specification seems pointless, it allows revealing weak points of the architecture, show that the measures that have been conceived are suitable to cover those weak points, or to what extent trade-offs have to be made due to inherent vulnerabilities. The protocol implemented by the framework CWA resembles only the most basic of the three protocols proposed by the DP-3T project. Hence, a formal verified model that stresses the impact or limits of certain security measures might give more weight to appeals like [19] to adopt the more sophisticated protocols.

The contributions of this paper are (a) formal re-engineering of CWA (b) providing an additional security and privacy analysis with interactive theorem proving certification of a novel view on the system architecture including actors, locations, and policies, (c) a formal definition of a security refinement process that allows improving a system based on attacks found by the attack tree analysis and (d) an application of the refinement to improve security of the CWA specification.

In this paper, we first provide some background in Section 2: we give a brief overview of related works and the protocol of CWA (Section 2.1). We then introduce the Isabelle Infrastructure framework (Section 2.2). Next, we present our model (Section 3) and analysis of privacy and attacks (Section 3.4). The found attack on the first abstract specification motivates refinement. The formal definition of refinement for the Isabelle Infrastructure framework is introduced and illustrated on CWA (Section 4) before drawing some conclusions (Section 5).

The formal model in the Isabelle insider framework is fully mechanized and proved in Isabelle (sources available [8]).

## 2 Background and related work

### 2.1 DP-3T and PEPP-PT

We are mainly concerned with the architecture and protocols proposed by the DP-3T (*Decentralized Privacy-Preserving Proximity Tracing*) project [16]. The main reason to focus on this particular family of protocols is the *Exposure Notification Framework* (ENF), jointly published by Apple and Google [1], that adopts core principles of the DP-3T proposal. This API is not only used in CWA but has the potential of being widely adopted in future app developments that might emerge due to the reach of players like Apple and Google.

There are, however, competing architectures noteworthy, namely protocols developed under the roof of the *Pan-European Privacy-Preserving Proximity Tracing* project (PEPP-PT) [21], e.g. PEPP-PT-ROBERT [22], that might be characterized as centralized architectures.

Neither DP-3T nor PEPP-PT are synonymous for just one single protocol. Each project endorses different protocols with unique properties in terms of privacy and data protection.

There is a variety of noteworthy privacy and security issues. The debate among advocates of centralized architectures and those in favor of a decentralized approach in particular yields a lot of interesting material detailing different attacks and possible mitigation strategies: [23], [20], [18].

In terms of attack scenarios, we focus on, what might be classified as deanonymization attacks: Tracking a device (see [18][p9], [23][p8]) and identifying infected individuals ([18][p5], [23][p9]).

**Basic DP-3T architecture** Upon installation, the app generates secret daily seeds to derive so-called *Ephemeral Ids* (EphIDs) from them. EphIDs are generated locally with cryptographic methods and cannot be connected to one another but only reconstructed from the secret seed they were derived from.

During normal operation each client broadcasts its EphIDs via Bluetooth whilst scanning for EphIDs broadcasted by other devices in the vicinity. Collected EphIDs are stored locally along with associated meta-data such as signal attenuation and date. In DP-3T the contact information gathered is never shared but only evaluated locally.

If patients test positive for the Corona-virus, they are entitled to upload specific data to a central backend server. This data is aggregated by the backend server and redistributed to all clients regularly to provide the means for local risk scoring, i. e., determining whether collected EphIDs match those broadcasted by now-confirmed Corona-virus patients during the last, e. g., 14, days.

In the most simple (and insecure) protocol proposed by DP-3T this basically translates into publishing the daily seeds used to derive EphIDs from. The protocol implemented by ENF and, hence, CWA adopts this low-cost design [15]. DP-3T proposes two other, more sophisticated protocols that improve privacy and data protection properties to different degrees but are more costly to implement. Figure 1 illustrates the basic system architecture along with some of the mitigation measures either implemented in CWA or proposed by DP-3T.

## 2.2 Isabelle Infrastructure framework

The Isabelle Infrastructure is built in the interactive generic theorem prover Isabelle/HOL [13]. As a framework, it supports formalization and proof of systems with actors and policies. It originally emerged from verification of insider threat scenarios but it soon became clear that the theoretical concepts, like temporal logic combined with Kripke structures and a generic notion of state transitions




Fig. 1. Decentralized privacy-preserving proximity tracing protocol of CWA

were very suitable to be combined with attack trees into a formal security engineering process [3] and framework [7]. Figure 2 gives an overview of the Isabelle Infrastructure framework with its layers of object-logics – each level below embeds the one above showing the novel contribution of this paper in blue on the top.

**Kripke structures, CTL, and Attack Trees** The Isabelle framework has now after various case studies become a general framework for the state-based security analysis of infrastructures with policies and actors. Temporal logic and Kripke structures build the foundation. Meta-theoretical results have been established to show equivalence between attack trees and CTL statements [4]. This foundation provides a generic notion of state transition on which attack trees and temporal logic can be used to express properties. The main notions used in this paper are:

- Kripke structures and state transitions:  
Using a generic state transition relation  $\rightarrow$ , Kripke structures are defined as a set of states reachable by  $\rightarrow$  from an initial state set, for example,


$$\text{Kripke } \{t. \exists i \in I. i \rightarrow^* t\} I$$

where  $\rightarrow^*$  is the reflexive transitive closure over  $\rightarrow$ .

- CTL statements:  
For example, we can write

$$K \vdash EF s$$

to express that in Kripke structure  $K$  there is a path on which the property  $s$  (a set of states) will eventually hold.



**Fig. 2.** Generic Isabelle Infrastructure framework applied to CWA.

- Attack trees:

The datatype of attack trees has three constructors:  $\oplus_{\vee}$  creates or-trees and  $\oplus_{\wedge}$  creates and-trees. And-attack trees  $l\oplus_{\wedge}^s$  and or-attack trees  $l\oplus_{\vee}^s$  consist of a list of sub-attacks – again attack trees. The third constructor creates a base attack as a pair of state sets written  $\mathcal{N}_{(I,s)}$ . For example, a two step and-attack leading from state set  $I$  via  $si$  to  $s$  is expressed as

$$\vdash [\mathcal{N}_{(I,si)}, \mathcal{N}_{(si,s)}] \oplus_{\wedge}^{(I,s)}$$

- Attack tree refinement, validity and adequacy:

Attack trees have their own refinement (not to be mixed up with the system refinement presented in this paper as introduced in the next section). An abstract attack tree  $A$  of type  $\alpha$  **attree** over an arbitrary state type  $\sigma$  may be refined by spelling out the attack steps until a valid attack is reached:

$$\vdash A :: (\sigma :: \text{state}) \text{ attree.}$$

The validity is defined constructively (code is generated from it) and its adequacy with respect to a formal semantics in CTL is proved and can be used to facilitate actual application verification as demonstrated here in the stepwise system refinements.

**Instantiation of Framework** The formal model of CWA uses the Isabelle Infrastructure framework instantiating it by reusing its concept of *actors* for users and smartphones whereby locations correspond to physical locations. The Ephemeral Ids, their sending and change is added to infrastructures by slightly adapting the basic state type of infrastructure graphs and accordingly the semantic rules for the actions move, get, and put. The details of the newly adapted infrastructure are presented in Section 3. Technically, an Isabelle theory file **Infrastructure.thy** builds on top of the theories for Kripke structures and CTL (**MC.thy**), attack trees (**AT.thy**), and security refinement (**Refinement.thy**).

Thus all these concepts can be used to specify the formal model for CWA, express relevant and interesting properties, and conduct interactive proofs (with the full support of the powerful and highly automated proof support of Isabelle). All Isabelle sources are available online [8].

**Refinement** An additional feature that has been integrated into the Isabelle Infrastructure framework motivated by security engineering formal specifications for IoT healthcare system is an extension of the formal specification process introducing refinement of Kripke structures [7,11]. It refines a system model based on a formal definition of a combination of trace refinement and structural refinement (or datatype refinement). The definition allows proving property preservation results crucial for an iterative development process. The refinements of the system specification can be interleaved with attack analysis while security properties can be proved in Isabelle. In each iteration security qualities are accumulated while continuously attack trees scrutinize the design. One of the contributions of this paper is to explore different concepts of refinement: the formal expression of refinement, enables to pin down (i.e. exemplify) different concepts of refinement (data refinement, action refinement, trace refinement (aka spec refinement) and combinations thereof with concrete attack scenarios.

### 3 Modeling and analyzing CWA

#### 3.1 Infrastructures, Policies, and Actors

The Isabelle Infrastructure framework supports the representation of infrastructures as graphs with actors and policies attached to nodes. These infrastructures are the *states* of the Kripke structure.

The transition between states is triggered by non-parameterized actions `get`, `move`, and `put` executed by actors. Actors are given by an abstract type `actor` and a function `Actor` that creates elements of that type from identities (of type `string` written `''s''` in Isabelle). Actors are contained in an infrastructure graph type `igraph` constructed by its constructor `Lgraph`.

```
datatype igraph =
  Lgraph (location × location)set
          location ⇒ identity set
          identity ⇒ (string set × string set × efid)
          location ⇒ string × (d1m × data) set
          location ⇒ efid set
          actor ⇒ location ⇒ (identity × efid) set
```

In the current application of the framework to the CWA case study, this graph contains a set of location pairs representing the topology of the infrastructure as a graph of nodes and a function that assigns a set of actor identities to each node (location) in the graph. The third component of an `igraph` assigns the credentials to each actor: a triple-valued function whose first range component

is a set describing the credentials in the possession of an actor and the second component is a set defining the roles the actor can take on; most prominently the third component is the **efid** assigned to the actor. This is initially just a natural number of type **nat** mapped by the constructor **Efid** to the new datatype **efid** but will be refined to actually represent lists of Ephemeral Ids later when refining the specification.

```
datatype efid = EfId nat
```

The fourth component of the type **igraph** assigns security labeled data to locations, a feature not used in the current application. The second to last component assigns the set of efids of all currently present smart phones to each location of the graph. The last component finally denotes the knowledge set of each actor for each location: a set of pairs of actors and potential ids.

Corresponding projection functions for each of the components of an infrastructure graph are provided; they are named **gra** for the actual set of pairs of locations, **agra** for the actor map, **cgra** for the credentials, **lgra** for the data at that location, **egra** for the assignment of current efids to locations, and **kgra** for the knowledge set for each actor for each location.

In CWA, the initial values for the **igraph** components use two locations **pub** and **shop** to define the following constants (we omit the data map component **ex\_locs**).

```
ex_loc_ass ≡ (λ x. if x = pub then {'Alice'', ''Bob'', ''Eve''}
                      else (if x = shop then {'Charly'', ''David''}
                             else {}))
ex_creds ≡ (λ x. if x = ''Alice'' then ({}, {}, EfId 1) else
              (if x = ''Bob'' then  ({}, {}, EfId 2) else
               (if x = ''Charly'' then ({}, {}, EfId 3) else
                (if x = ''David'' then ({}, {}, EfId 4) else
                 (if x = ''Eve'' then ({}, {}, EfId 5)
                  else ({}, {}, EfId 0))))))
ex_efids ≡ (λ x. if x = pub then {EfId 1, EfId 2, EfId 5}
              else (if x = shop then {EfId 3, EfId 4} else {}))
ex_knos ≡ (λ x. (if x = Actor ''Eve'' then (λ l. {}) else (λ l. {})))
```

These components are wrapped up into the following **igraph**.

```
ex_graph ≡
  Lgraph {(pub, shop)} ex_loc_ass ex_creds ex_locs ex_efids ex_knos
```

Infrastructures are given by the following datatype that contains an infrastructure graph of type **igraph** and a policy given by a function that assigns local policies over a graph to all locations of the graph.

```
datatype infrastructure = Infrastructure igraph
                               [igraph, location] ⇒ policy set
```

There are projection functions **graphI** and **delta** when applied to an infrastructure return the graph and the local policies, respectively. The function

`local_policies` gives the policy for each location `x` over an infrastructure graph `G` as a pair: the first element of this pair is a function specifying the actors `y` that are entitled to perform the actions specified in the set which is the second element of that pair. The local policies definition for CWA, simply permits all actions to all actors in both locations.

```
local_policies G ≡
  (λ x. if x = pub then {(λ y. True, {get,move,put})}
    else (if x = shop then {(λ y. True, {get,move,put})} else {}))
```

For CWA, the initial infrastructure contains the graph `ex_graph` with its two locations `pub` and `shop` and is then wrapped up with the local policies into the infrastructure `Corona_scenario` that represents the “initial” state for the Kripke structure.

```
Corona_scenario ≡ Infrastructure ex_graph local_policies
```

### 3.2 Policies, privacy, and behaviour

Policies specify the expected behaviour of actors of an infrastructure. They are given by pairs of predicates (conditions) and sets of (enabled) actions. They are defined by the `enables` predicate: an actor `h` is enabled to perform an action `a` in infrastructure `I`, at location `l` if there exists a pair `(p,e)` in the local policy of `l` (`delta I l` projects to the local policy) such that the action `a` is a member of the action set `e` and the policy predicate `p` holds for actor `h`.

```
enables I l h a ≡ ∃ (p,e) ∈ delta I l. a ∈ e ∧ p h
```

The Privacy protection goal is to avoid deanonymization. That is, an attacker should not be able to disambiguate the set of pairs of real ids and EphIDs. This is abstractly expressed in the predicate `identifiable`.

```
identifiable eid A ≡ is_singleton{(Id,Eid). (Id, Eid) ∈ A ∧ Eid = eid}
```

The predicate `identifiable` is used to express as the global policy ‘Eve cannot deanonymize an Ephemeral Id `eid` using the gathered knowledge’:

```
global_policy I eid ≡
  ¬(identifiable eid
    ((λ (kgra(graphI I) (Actor ''Eve'') (nodes(graphI I))))
     - {(x,y). x = ''Eve''}))
```

### 3.3 Infrastructure state transition

The state transition relation uses the syntactic infix notation  $I \rightarrow I'$  to denote that infrastructures `I` and `I'` are in this relation. We will use this state transition to define the dynamic behaviour of the system including actors’ movements and the evolution of their knowledge. This allows representing attacks as sequences of state transitions leading to states in which the global policy is violated. To



give an intuitive motivating idea before the formal presentation in Section 3.4: the attack will consist of the state transitions of Eve getting available EphIds at the location pub, Alice moving to shop, Eve following her and again getting the EphIds at shop.

So, one relevant rule for the state transition is the one for `move` that defines the state transition relation expressing state changes when actors move across the infrastructure.

```

move:  $G = \text{graphI } I \implies a @_G l \implies a \in \text{actors\_graph}(\text{graphI } I) \implies$ 
 $l \in \text{nodes } G \implies l' \in \text{nodes } G \implies \text{enables } I \ l' \ (\text{Actor } h) \ \text{move} \implies$ 
 $I' = \text{Infrastructure } (\text{move\_graph\_a } a \ l \ l' \ (\text{graphI } I))(\text{delta } I)$ 
 $\implies I \rightarrow I'$ 

```

The semantics of this rule is embedded in the function `move_graph_a` that adapts the infrastructure state so that the moving actor `a` is now associated to the target location `l'` in the actor map `agra` and not any more at `l` and also the association of `efids` is updated accordingly.

```

move_graph_a  $n \ l \ l' \ g \equiv$ 
  Lgraph (gra  $g$ )
    (if  $n \in ((\text{agra } g) \ l) \wedge n \notin ((\text{agra } g) \ l')$  then
       $((\text{agra } g)(l := (\text{agra } g \ l) - \{n\}))(l' := (\text{insert } n \ (\text{agra } g \ l')))$ 
    else  $(\text{agra } g)$ )
    (cgra  $g$ )(lgra  $g$ )
    (if  $n \in ((\text{agra } g) \ l) \wedge n \notin ((\text{agra } g) \ l')$  then
       $((\text{egra } g)(l := (\text{egra } g \ l) - \{\text{efemid } (\text{cgra } g \ n)\}))$ 
       $(l' := (\text{insert } (\text{efemid } (\text{cgra } g \ n))(\text{egra } g \ l')))$ 
    else  $\text{egra } g$ )
    (kgra  $g$ )

```

Another action relevant for the deanonymization attack describes how an actor gets information. This is defined by the rule of the state transition for the action `get`. Initially, this rule expresses that an actor that resides at a location `l` ( $a @_G l$ ) and is enabled by the local policy in this location to “get” can combine all ids at the current location (contained in `egra G l`) with all actors at the current location (contained in `agra G l`) and add this set of pairs to his knowledge set `kgra G` using the function `update f(l := n)` redefining the function `f` for the input `l` to have now the new value `n`.

```

get_data:  $G = \text{graphI } I \implies a @_G l \implies \text{enables } I \ l \ (\text{Actor } a) \ \text{get} \implies$ 
 $I' = \text{Infrastructure}$ 
  (Lgraph (gra  $G$ )(agra  $G$ )(cgra  $G$ )(lgra  $G$ )(egra  $G$ )
     $((\text{kgra } G)((\text{Actor } a) := ((\text{kgra } G \ (\text{Actor } a))(l :=$ 
       $\{(x,y). x \in \text{agra } G \ l \wedge y \in \text{egra } G \ l\}))))$ 
    (delta  $I$ )
   $\implies I \rightarrow I'$ 

```

Based on this state transition and the above defined `Corona_scenario`, we define the first Kripke structure.

```

corona_Kripke  $\equiv \text{Kripke } \{ I. \text{Corona\_scenario} \rightarrow^* I \} \{ \text{Corona\_scenario} \}$ 

```

### 3.4 Attack analysis

For the analysis of attacks, we negate the security property that we want to achieve, usually the global policy.

Since we consider a predicate transformer semantics, we use sets of states to represent properties. The invalidated global policy is given by the set `scorona`.

$$\text{scorona} \equiv \{x. \exists n. \neg \text{global\_policy } x \text{ (Efid } n)\}$$

The attack we are interested in is to see whether for the scenario

$$\text{Kripke\_scenario} \equiv \text{Infrastructure ex\_graph local\_policies}$$

from the initial state  $\text{Icorona} \equiv \{\text{corona\_scenario}\}$ , the critical state `scorona` can be reached, that is, is there a valid attack  $(\text{Icorona}, \text{scorona})$ ?

For the Kripke structure `corona.Kripke` we first derive a valid and-attack using the attack tree proof calculus.

$$\vdash [\mathcal{N}_{(\text{Icorona}, \text{Corona})}, \mathcal{N}_{(\text{Corona}, \text{Corona1})}, \mathcal{N}_{(\text{Corona1}, \text{Corona2})}, \mathcal{N}_{(\text{Corona2}, \text{Corona3})}, \mathcal{N}_{(\text{Corona3}, \text{scorona})}] \oplus_{\wedge}^{(\text{Icorona}, \text{scorona})}$$

The sets `Corona`, `Corona1`, `Corona2`, `Corona3` are the intermediate states where **Bob** moves to shop and **Eve** follows him collecting the Ephemeral Ids in each location. The collected information enables identifying Bob's Ephemeral Id.

The attack tree calculus [4] exhibits that an attack is possible.

$$\text{corona\_Kripke} \vdash \text{EF scorona}$$

We can simply apply the Correctness theorem `AT_EF` [4,5]

$$\begin{aligned} &\vdash A \implies \text{attack } A = (I, s) \\ &\implies \text{Kripke } \{s \mid \exists i \in I. (i \rightarrow^* s)\} \vdash \text{EF } s \end{aligned}$$

to immediately prove this CTL statement. This application of the meta-theorem of Correctness of attack trees saves us proving the CTL formula tediously by exploring the state space in Isabelle proofs. Alternatively, we could use the generated code for the function `is_attack_tree` in Scala (see [4]) to check that a refined attack of the above is valid.

## 4 Refinement

Intuitively, a refinement changes some aspect of the type of the state, for example, replaces a data type by a richer datatype or restricts the behaviour of the actors. The former is expressed directly by a mapping of datatypes, the latter is incorporated into the state transition relation of the Kripke structure that corresponds to the transformed model. In other words, we can encode a refinement within our framework as a relation on Kripke structures that is parameterized additionally by a function that maps the refined type to the abstract type. The direction “from refined to abstract” of this type mapping may seem curiously counter-intuitive. However, the actual refinement is given by the refinement that

uses this function as an input. The refinement then refines an abstract to a more concrete system specification. The additional layer for the refinement can be formalized in Isabelle as a refinement relation  $\sqsubseteq_{\mathcal{E}}$ . This relation, called **refinement**, is typed as a relation over triples – a function from a threefold Cartesian product to **bool**, the type containing true and false only. The type variables  $\sigma$  and  $\sigma'$  input to the type constructor **Kripke** represent the abstract state type and the concrete state type. Consequently, the middle element of the triples selected by the relation **refinement** is a function of type  $\sigma' \Rightarrow \sigma$  mapping elements of the refined state to the abstract state. The expression in quotation marks after the type is again the infix syntax in Isabelle that allows the definition of mathematical notation instead of writing **refinement** in prefix manner. This nicer infix syntax is already used in the actual definition. Finally, the arrow  $\Longrightarrow$  is the implication of Isabelle’s meta-logic while  $\longrightarrow$  is the one of the *object* logic HOL. They are logically equivalent but of different types: within a HOL formula, for example, within the below  $\forall s' \in \text{states } K' \dots$ , only the implication  $\longrightarrow$  can be used.

```

refinement :: ( $\sigma$  Kripke  $\times$  ( $\sigma' \Rightarrow \sigma$ )  $\times$   $\sigma'$  Kripke)  $\Rightarrow$  bool (" $\_ \sqsubseteq_{\mathcal{E}} \_$ ")
 $K \sqsubseteq_{\mathcal{E}} K' \equiv \forall s' \in \text{states } K'. \forall s \in \text{init } K'.
    s \rightarrow_{\sigma'}^* s' \longrightarrow \mathcal{E}(s) \in \text{init } K \wedge \mathcal{E}(s) \rightarrow_{\sigma}^* \mathcal{E}(s')$ 

```

The definition of  $K \sqsubseteq_{\mathcal{E}} K'$  states that for any state  $s'$  of the refined Kripke structure that can be reached by the state transition in zero or more steps from an initial state  $s$  of the refined Kripke structure, the mapping  $\mathcal{E}$  from the refined to the abstract model’s state must preserve this reachability, i.e., the image of  $s$  must also be an initial state and from there the image of  $s'$  under  $\mathcal{E}$  must be reached within 0 and  $n$  steps.

#### 4.1 Property Preserving System Refinement

A first direct consequence of this definition is the following lemma where the operator ‘ in  $\mathcal{E}'(\text{init } K')$  represents function image, i.e.,  $\{\mathcal{E}(x) \mid x \in \text{init } K'\}$ .

**lemma** **init\_ref**:  $K \sqsubseteq_{\mathcal{E}} K' \Longrightarrow \mathcal{E}'(\text{init } K') \subseteq \text{init } K$

A more prominent consequence of the definition of refinement is that of property preservation. Here, we show that refinement preserves the CTL property of **EFs** which means that a reachability property true in the refined model  $K'$  is already true in the abstract model. A state set  $s'$  represents a property in the predicate transformer view of properties as sets of states. The additional condition on initial states ensures that we cannot “forget” them.

**theorem** **prop\_pres**:

```

 $K \sqsubseteq_{\mathcal{E}} K' \Longrightarrow \text{init } K \subseteq \mathcal{E}'(\text{init } K') \Longrightarrow$ 
 $\forall s' \in \text{Pow}(\text{states } K'). K' \vdash \text{EF } s' \longrightarrow K \vdash \text{EF } (\mathcal{E}'(s'))$ 

```

It is remarkable, that our definition of refinement by Kripke structures entails property preservation and makes it possible to prove this as a theorem in Isabelle once for all, i.e., as a meta-theorem. However, this is due to the fact that

our generic definition of state transition allows explicitly formalizing such sophisticated concepts like reachability. For practical purposes, however, the proof obligation of showing that a specific refinement is in fact a refinement is rather complex justly because of the explicit use of the transitive closure of the state transition relation. In most cases, the refinement will be simpler. Therefore, we offer additional help by the following theorem that uses a stronger characterization of Kripke structure refinement and shows that our refinement follows from this.

**theorem strong\_mt:**

$$\begin{aligned} \mathcal{E}'(\text{init } K') &\subseteq \text{init } K \wedge s \rightarrow_{\sigma'} s' \longrightarrow \mathcal{E}(s) \rightarrow_{\sigma} \mathcal{E}(s') \\ \implies K &\sqsubseteq_{\mathcal{E}} K' \end{aligned}$$

This simpler characterization is in fact a stronger one: we could have  $s \rightarrow_{\sigma'} s'$  in the refined Kripke structure  $K'$  and  $\neg(\mathcal{E}(s) \rightarrow_{\sigma} \mathcal{E}(s'))$  but neither  $s$  nor  $s'$  are reachable from initial states in  $K'$ . For cases, where we have the simpler one-step proviso and also reachability from initial states we provide a slightly weaker version of **strong\_mt**.

**theorem strong\_mt':**

$$\begin{aligned} \mathcal{E}'(\text{init } K') &\subseteq \text{init } K \wedge (\exists s_0 \in \text{init } K'. s_0 \rightarrow^* s) \\ \wedge s \rightarrow_{\sigma'} s' &\longrightarrow \mathcal{E}(s) \rightarrow_{\sigma} \mathcal{E}(s') \implies K \sqsubseteq_{\mathcal{E}} K' \end{aligned}$$

This idea of property preservation coincides with the classical idea of trace refinement as it is given in process algebras like CSP. In this view, the properties of a system are given by the set of its event traces. Now, a refinement of the system is given by another system that has a subset of the event traces of the former one. Although the principal idea is similar, we greatly extend it since our notion additionally incorporates data refinement. Since we include a state map  $\sigma' \Rightarrow \sigma$  in our refinement map, we additionally allow structural refinement: the state map generalizes the basic idea of trace refinement by traces corresponding to each other but allows additionally an exchange of data types. As we see in the application to the case study, the refinement steps may sometimes just specialize the traces: in this case the state map  $\sigma' \Rightarrow \sigma$  is just identity.

In addition, we also have a simple implicit version of *action refinement*. In an action refinement, traces may be refined by combining consecutive system events into atomic events thereby reducing traces. We can observe this kind of refinement in the second refinement step of CWA considered next.

## 4.2 Refining the Specification

Clearly, fixed Ephemeral Ids are not really ephemeral. The model presented in Section 3 has deliberately been designed abstractly to allow focusing on the basic system architecture and finding an initial deanonymization attack. We now introduce “proper” Ephemeral Ids and show how the system datatype can be refined to a system that uses those instead of the fixed ones.

For the DP-3T Ephemeral Ids [17], for each day  $t$  a seed  $SK_t$  is used to generate a list of length  $n = 24 * 60 / L$ , where  $L$  is the duration for which the Ephemeral Ids are posted by the smart phone

`EphID1 || ... || EphIDn = PRG(PRF(SKt, 'broadcast key'))`

“where PRF is a pseudo-random function (e.g., HMAC-SHA256), “broadcast key” is a fixed, public string, and PRG is a pseudorandom generator (e.g. AES in counter mode)” [17].

From a cryptographic point of view, the crucial properties of the Ephemeral Ids are that they are purely random, therefore, they cannot be guessed, but, at the same time, if – after the actual encounter between sender and receiver – the seed  $SK_t$  is published, it is feasible to relate any of the  $EphID_i$  to  $SK_t$  for all  $i \in \{1, \dots, n\}$ . For a formalization of this crucial cryptographic property in Isabelle it suffices to define a new type of list of Ephemeral Ids **efidlist** containing the root  $SK_t$  (the first **efid**), a current **efid** indicated by a list pointer of type **nat**, and the actual list of **efids**.

**datatype** **efidlist** = **Efids** "efid" "nat" "efid list"

We define functions for this datatype: **efidsroot** returning the first of the three constituents in an **efidlist** (the root  $SK_t$  of type **efid**); **efids\_index** giving the second component of type **nat**, that is, the index of the current **efid**; **efids\_inc\_ind** applied to an **efidlist** increments the index; **efids\_cur** returning the current **efid** from the list and **efids\_list** for the entire list (the third component of type **efid list**).

The first step of refinement replaces the simple **efid** in the infrastructure type by the new type **efidlist**. Note, that in the new datatype **igraph** this change affects only the third component, the credentials to become

**identity**  $\Rightarrow$  (**string set**  $\times$  **string set**  $\times$  **efidlist**)

The last two components, the set of currently present **efids**, and the knowledge set, remain the same and still operate on the simple type **efid**. The refined state transition relation implements the possibility of changing the Ephemeral Ids by the rule for the action **put** that resembles very much the rule for **get**. The important change to the infrastructure state is implemented in the function **put\_graph\_efid** that increases the current index in the **efidlist** in the credential component **cgra g n** for the “putting” actor identity **n** and inserts the current **efid** from that credential component into the **egra** component, the set of currently “present” Ephemeral Ids at the location **l**.

```
put_graph_efid n l g  $\equiv$ 
  Lgraph (gra g)(agra g)
    ((cgra g)(n := (credentials (cgra g n), roles (cgra g n),
      efids_inc_ind(efemid (cgra g n)))))
    (lgra g)
    ((egra g)(l := insert (efids_cur(efemid (cgra g n)))(egra g l)))
    (kgra g)
```

We can now apply the refinement by defining a datatype map from the refined infrastructure type **InfrastructureOne.infrastructure** to the former one **Infrastructure.infrastructure**.

```

definition refmap :: InfrastructureOne.infrastructure  $\Rightarrow$ 
    Infrastructure.infrastructure
where refmap I =
    Infrastructure.Infrastructure
      (Infrastructure.Lgraph
        (InfrastructureOne.gra (graphI I))
        (InfrastructureOne.agra (graphI I))
        ( $\lambda$  h. repl_efr
          (InfrastructureOne.cgra (graphI I)) h)
          (InfrastructureOne.lgra (graphI I))
          (InfrastructureOne.egra (graphI I))
          ( $\lambda$  a.  $\lambda$  l.
            ( $\lambda$  (x,y).(x, efids_root(efemid(InfrastructureOne.cgra (graphI I) x))))
            ' (InfrastructureOne.kgra (graphI I)) a l)))

```

This is then plugged into the parameter  $\mathcal{E}$  of the refinement operator allowing to prove `corona_Kripke`  $\sqsubseteq_{\text{refmap}}$  `corona_Kripke0` where the latter is the refined Kripke structure.

Surprisingly, we can still prove `corona_Kripke0`  $\vdash$  EF `scorona0` by using the same attack tree as in the abstract model: if Bob moves from pub to shop<sup>1</sup>, he is vulnerable to being identifiable as long as he does not change the current `efid`. So, if Eve moves to the shop as well and performs a `get` before Bob does a `put`, then Eve's knowledge sets permit identifying Bob's current Ephemeral Id as his.

**Second refinement step** The persistent attack can be abbreviated informally by the action sequence [`get`,`move`,`move`,`get`] performed by actors Eve, Bob, Eve, and Eve again, respectively. How can a second refinement step avoid that Eve does the last `get` by imposing that after the first `move` of Bob a `put` action must happen before Eve can do another `get`? A very simple remedy to exclude this attack is to bind a `put` action after every `move`. We can implement that change by a minimal update to the function `move_graph_a` (see Section 3) by adding an increment (highlighted in the code snippet in bold letters) of the currently used Ephemeral Id before updating the `egra` component of the target location.

```

move_graph_a n l l' g  $\equiv$  ...
  (l' := insert (efids_cur(efids_inc_ind(efemid (cgra g n)))) (egra g l))...

```

This is an action refinement because the `move` action is changed. It is a refinement, since any trace of the refined model can still be mapped to a trace in the more abstract model just omitting a few steps (the refinement relation is defined using the reflexive transitive closure  $\rightarrow^*$ ).

<sup>1</sup> that is, if he moves alone: if all others from pub go to shop with him his anonymity remains intact.

## 5 Summary and discussion of relevance of the approach

We can establish in our formal framework an attack that even a system using changing Ephemeral ids can be broken if the attacker physically follows a victim. This is a basic attack on anonymity: a user's connection between his Ephemeral Ids and personal details (Iphone MAC or name) is revealed to the attacker. The protection goal of privacy is thereby destroyed.

When establishing the attack we start from a simplistic scenario that does not use Ephemeral Ids but fixed ids. In this (over)simplified model the attack is established. We then define a formal refinement calculus for the Isabelle Infrastructure framework to refine the attack to a system with Ephemeral Ids that change in fixed time intervals obfuscating the relationship between user and his pseudonym<sup>2</sup>.

Now, the refinement shows that although the Ephemeral Ids change regularly the same attack that has been identified on the very abstract level (fixed ids) persists. The refinement allows refining the datatype ( $\text{Id} \mapsto \text{EphID}$ ) but also delivers the usual trace refinement (behaviours of the refined systems are a subset of the traces of the abstract system). This persistence of the attack precisely shows which part of the system behaviour is responsible for the attack. In a second refinement step using action refinement based on this insight from the (repeated) attack, we can exclude the dangerous attack trace.

Clearly, the abstract attack we establish is obvious on an informal level but the persistence of the attack on the refined system is less obvious. The remedy by a second refinement step is an evident restriction of the system behaviour which gives a clear specification of a system secured against this attack. The use of formal methods therefore lies not in the discovery of an obvious attack on a simplified system but in showing how a formal specification including security refinement can lead to a stepwise improvement that is accompanied by formal proof in the Isabelle Infrastructure framework. The solution to exclude the attack in the second refinement step binds the action move together with a put action. This shows that besides datatype refinement and trace refinement our refinement calculus also entails action refinement. This action refinement is implemented implicitly by changing the effects of the actions in the semantic state transition relation. In future work, we could think about making the action refinement more explicit by considering a relationship between semantic rules or by refining the refinement notion to a more explicit layer of protocol steps – similar to what has been done in previous applications of the Isabelle Infrastructure framework for example to Auction protocols [9] or the Quantum Key Distribution [6].

The security refinement in general might seem pointless, as in the first step of refinement the attack persists and even though the second refinement gets rid of this specific attack, it doesn't exclude the reachability of the attack goal altogether (if the attacker Eve gets Bob on his own in a location she can map all used EphIDs to him). However, the refinement makes the system relatively

---

<sup>2</sup> We identify the smartphone and the user which might be also recognized by his appearance (face)

more secure in that for a larger number of traces the abstract attack does not work anymore<sup>3</sup>. It is important to emphasize that security refinement is a cyclic process that improves security but does not usually terminate (like a loop) with a fixed point of 100% secure system. In a refined model new detail may give rise to new attack possibilities. These additional attacks can be identified using the Attack Tree calculus and trigger further refinement steps.

## References

1. Apple and Google. Exposure notification framework, 2020. <https://www.google.com/covid19/exposurenotifications/>.
2. D. Bundesregierung. Die Corona-Warn-App: Unterstützt uns im Kampf gegen Corona, 2020. German government announcement and support of Coronavirus warning app, <https://www.bundesregierung.de/breg-de/themen/corona-warn-app>.
3. CHIST-ERA. Success: Secure accessibility for the internet of things, 2016. <http://www.chistera.eu/projects/success> and <https://github.com/success-iot>.
4. F. Kammüller. Attack trees in Isabelle. In *20th International Conference on Information and Communications Security, ICICS2018*, volume 11149 of *LNCS*. Springer, 2018.
5. F. Kammüller. Formal modeling and analysis of data protection for GDPR compliance of IoT healthcare systems. In *IEEE Systems, Man and Cybernetics, SMC2018*. IEEE, 2018.
6. F. Kammüller. Attack trees in Isabelle extended with probabilities for quantum cryptography. *Computers & Security*, 87, 2019.
7. F. Kammüller. Combining secure system design with risk assessment for IoT healthcare systems. In *Workshop on Security, Privacy, and Trust in the IoT, SPTIoT’19, colocated with IEEE PerCom*. IEEE, 2019.
8. F. Kammüller. Isabelle infrastructure framework with IoT healthcare S&P application and corona-virus warn app, 2020. Available at <https://github.com/flokam/IsabelleAT>.
9. F. Kammüller, M. Kerber, and C. Probst. Insider threats for auctions: Formal modeling, proof, and certified code. *Journal of Wireless Mobile Networks, Ubiquitous Computing, and Dependable Applications (JoWUA)*, 8(1):44–78, 2017. Special Issue on Insider Threat Solutions - Moving from Concept to Reality.
10. F. Kammüller and U. Nestmann. Inter-blockchain protocols with the Isabelle Infrastructure framework. In *Formal Methods for Blockchain, 2nd Int. Workshop, colocated with CAV’20*, volume 84 of *Open Access series in Informatics*. Dagstuhl publishing, 2020.
11. F. Kammüller. A formal development cycle for security engineering in Isabelle, 2020. Cornell University, arxiv.org <https://arxiv.org/abs/2001.08983>.
12. F. Kammüller and M. Kerber. Applying the Isabelle Insider framework to air-plane security, 2020. Cornell University, arxiv.org <https://arxiv.org/abs/2003.11838>.
13. T. Nipkow, L. C. Paulson, and M. Wenzel. *Isabelle/HOL – A Proof Assistant for Higher-Order Logic*, volume 2283 of *LNCS*. Springer-Verlag, 2002.
14. The Corona-Warn-App Project, 2020. <https://github.com/corona-warn-app>.

<sup>3</sup> Adding probabilities as in [6] enables quantifying this.



15. The Corona-Warn-App Project. Corona-warn-app solution architecture, 2020. [https://github.com/corona-warn-app/cwa-documentation/blob/master/solution\\_architecture.md](https://github.com/corona-warn-app/cwa-documentation/blob/master/solution_architecture.md).
16. The DP-3T Project. Decentralized Privacy-Preserving Proximity Tracing, 2020. <https://github.com/DP-3T>.
17. The DP-3T Project. Decentralized privacy-preserving proximity tracing - White Paper, 2020. <https://github.com/DP-3T/documents/blob/master/DP3TWhitePaper.pdf>.
18. The DP-3T Project. Privacy and security risk evaluation of digital proximity tracing systems, 2020. <https://github.com/DP-3T/documents/blob/master/Securityanalysis/PrivacyandSecurityAttacksonDigitalProximityTracingSystems.pdf>.
19. The DP-3T Project. README: Apple / Google Exposure Notification, 2020. <https://github.com/DP-3T/documents>.
20. The DP-3T Project. Response to 'Analysis of DP3T: Between Scylla and Charybdis', 2020. <https://github.com/DP-3T/documents/blob/master/Securityanalysis/Responseto'AnalysisofDP3T'.pdf>.
21. The PEPP-PT Project. Pan-European Privacy-Preserving Proximity Tracing, 2020. <https://github.com/PEPP-PT>.
22. The ROBERT Project. ROBust and privacy-presERving proximity Tracing protocol, 2020. <https://github.com/ROBERT-proximity-tracing>.
23. S. Vaudenay. Analysis of DP3T: Between Scylla and Charybdis, 2020. <https://eprint.iacr.org/2020/399.pdf>.